
/*Kristin Sainani, March 18, 2018*/

/*Code for calculating Type I and Type II error for clinical MBI, following the equations given in the paper*/
%LET n=50; *sample size per group;
%LET ES=0; *true effect size;
%LET var=.364; *true variance—the specific value of 0.364 is from a pre-post design with r=.818. Std dev at baseline=1.0. Variance in change scores=1+1-2*.818=.364;
%LET deltab=0.2; *threshold for benefit;
%LET deltah=0.2; *threshold for harm;
%LET maxrisk=.005; *maximum risk of harm;
%LET minben=.25; *minimum risk of benefit;
data _null_;
df=2*&n-2;
err=sqrt(2)*sqrt(&var)/sqrt(&n); *standard error;
th=Tinv((1-&maxrisk),df); *T for harm, e.g. 2.68 for 99% CI;
tb=Tinv(1-&minben,df); *T for benefit, e.g. .68 for 50% CI;
hterm=-&deltah+th*err; 
bterm=&deltab-tb*err; 
maxterm=((df)*&n*(&deltab+&deltah)**2)/(2*&var*(th+tb)**2);
pharmmax=1-probchi(maxterm,(df)); *probability that the harm constraint is bigger;
pbenmax=1-pharmmax; *probability that the benefit constraint is bigger;
Tharm=(hterm-&ES)/err;
Tben=(bterm-&ES)/err;
if &ES<&deltab then do;
TypeI=pharmmax*(1-probT(Tharm,df))+pbenmax*(1-ProbT(Tben,df));
typeII=.;
end;
if &ES>=&deltab then do;
typeII=pharmmax*probT(Tharm, df)+pbenmax*ProbT(Tben,df);
typeI=.;
end;
putlog TypeI TypeII;
run;

/*Type I and II error curves by sample size, for comparison with simulations*/
%LET ES=0.2; *true effect size;
%LET var=.364; *true variance;
%LET deltab=0.2; *threshold for benefit;
%LET deltah=0.2; *threshold for harm;
%LET maxrisk=.005; *maximum risk of harm;
%LET minben=.25; *minimum risk of benefit;
data errors;
do n=10 to 150 by 10;
df=2*n-2;
err=sqrt(2)*sqrt(&var)/sqrt(n); *standard error;
th=Tinv((1-&maxrisk),df); *T for harm, e.g. 2.68 for 99% CI;
tb=Tinv(1-&minben,df); *T for benefit, e.g. .68 for 50% CI;
hterm=-&deltah+th*err; 
bterm=&deltab-tb*err; 
maxterm=((df)*n*(&deltab+&deltah)**2)/(2*&var*(th+tb)**2);
pharmmax=1-probchi(maxterm,(df));
pbenmax=1-pharmmax;
Tharm=(hterm-&ES)/err;
Tben=(bterm-&ES)/err;
TypeI=pharmmax*(1-probT(Tharm,df))+pbenmax*(1-ProbT(Tben,df));
typeII=pharmmax*probT(Tharm, df)+pbenmax*ProbT(Tben,df);
output;
end;
run;
/*Graph Type I or II error against sample size*/
goptions reset = all;
goptions htext=3;
axis1 label=(angle=90);
axis2 order=10 to 150 by 20;

proc gplot data=errors;
plot typeI*n/vaxis=axis1 haxis=axis2;
symbol1 v=dot c=red i=join h=2 w=2;
label N="Sample size per group";
label typeI="Type I error (%)";
run;

proc gplot data=errors;
plot typeII*n/vaxis=axis1 haxis=axis2;
symbol1 v=dot c=red i=join h=2 w=2;
label N="Sample size per group";
label typeII="Type II error (%)";
run;
